# One-Zone Sample Loaded in Shear with Strain Rate Reversal

Note

To view this project in FLAC3D, use the menu command Help ‣ Examples…. Choose “Dynamic/ ShearWithStrainRateReversal” and select “ShearWithStrainRateReversal.f3prj” to load. The main data files is shown at the end of this example. The remaining data files can be found in the project.

The operation of the various rules involved in the Hysteretic Damping Formulation is illustrated in this very simple example.

A simple one-zone model is fixed in all degrees-of-freedom, then shear strain is imposed on it by initializing the $$x$$-velocity of the top to a fixed value. This initial loading is interrupted by a small unload/load cycle; after this there is a complete unloading (extending to negative strain) followed by a loading part that continues to a higher positive strain level than before. In particular, note the half-scale initial loading curve, the slope of $$G_o$$ at each reversal, and the restoration of the original loading path after execution of the small loop. In this case, both the positive and negative stacks are popped upon closure of the small loop (i.e., the entire loop is forgotten), but only the information from the positive stack is used to restore state information; the negative-stack information is discarded.

Figure 1: Shear stress vs shear strain, with one reversal to show the effect of memory.

Data File

;-----------------------------------------------------------------------
;  Script file for dynamic problem
;  'One-zone sample loaded in shear with strain rate reversal'
;-----------------------------------------------------------------------
model new
model large-strain off
model title ...
"Shear stress vs shear strain, with 1 reversal to show the effect of memory"
model configure dynamic
; Create zones
zone create brick size 1 1 1
; Assign material model and properties
zone cmodel assign elastic
zone property density 1000 shear 5e8 bulk 10e8
; Boundary conditions
zone gridpoint fix velocity
zone gridpoint initialize velocity-x 1e-2 range position-z=1
; Dynamic setup
model dynamic timestep fix 1e-4
zone dynamic damping hysteretic default -3.5 1.3
history interval 1
; Histories
zone history stress-xz zoneid 1
zone history strain-increment-xz zoneid 1
; Cycle through several reversals
model cycle 1000
zone gridpoint initialize velocity-x -1 multiply
model cycle 250
zone gridpoint initialize velocity-x -1 multiply
model cycle 500
zone gridpoint initialize velocity-x -1 multiply
model cycle 2000
zone gridpoint initialize velocity-x -1 multiply
model cycle 2200